
OOP mit Pascal/Delphi

Grundlagen

Klassen erstellen

Die folgenden Übungen werden teils in Delphi als Konsolenanwendung erstellt („reines“ Pascal) und teils in „normalem“ Delphi.

Beispielaufgabe: Es ist eine Preisberechnung nach folgendem Schema objekt​orientiert zu codieren:

Konsolenanwendung

Delphi-Anwendung

Lösung (Konsolenanwendung):

Erklärungen:

(Hinweis: Aus Gründen der Vereinfachung erfolgt hier die Zeilennummerierung so, als stünden die Unit und das Hauptprogramm in einer Quelltextdatei, was na-türlich nicht der Fall ist.)

Die objektorientierte Programmierung ist keiner Programmiersprachen-Gene​ration zuzuordnen; viele Sprachen der dritten, vierten und fünften Generation bieten entsprechende Spracherweiterungen an. Wichtiges Kennzeichen einer objektorientierten Programmiersprache ist die Möglichkeit, Daten und Funk​tionen/Prozeduren derart zu kombinieren, dass die Daten nur mit definierten Funktionen manipulieret werden können. Dieser Mechanismus wird als Kapse​lung bezeichnet. In einem Klasse lassen sich so Datenstrukturen und Programm​quellcode zusammengefassten. Im folgenden Beispiel wurde dies von Zeile 4 bis 15 realisiert. Die Klasse (class) wird in einem Type-Block deklariert und bein​haltet zunächst einige Datenelemente ("preis", "menge" usw.); hier kann es sich also um eine Datenstruktur handeln, die in NICHT-OO-PASCAL mit "record" vereinbart werden könnte. In den Zeilen 12-14 erfolgt die Deklaration von so​genannten Methodenfunktionen. Eine "normale" record-Vereinbarung würde dies nicht zulassen! Implementiert werden die Methoden dann im Implemen​tations-Abschnitt (ab Zeile 18). Die Verknüpfung zwischen der externen Funk​tion und der Klasse erfolgt durch das Voranstellen des Klassennamens vor den Methodennamen, wobei beide Namen mit einem „.“ getrennt werden (Zeilen 18, 23, 32). Der Zugriff auf die einzelnen Datenelemente (von außen) lässt sich kontrollieren. Die Schlüsselwörter "privat" und "public" deklarieren in den Zeilen 6 und 10 die class-Elemente entsprechend. 

So ist ein direkter Zugriff auf das Element "preis" aus dem Hauptprogramm nicht erlaubt, z. B.:


p.preis:=50;

Der Compiler würde mit einer entsprechenden Fehlermeldung reagieren. 

Das "public"-Element "proz" darf auch im Hauptprogramm manipuliert werden:


p.proz:=3;

Auch hier sei darauf hingewiesen, dass eine "normale" record-Vereinbarung eine solche Differenzierung der Zugriffsmöglichkeiten auf die Record-Elemente nicht zulässt. Die Elemente dürfen eben nur mit den zugelassenen Funktionen verändert werden (Kapselung!). Das Hauptprogramm greift nun mit dem "." - Operator gleichermaßen auf Daten (z. B.: p.proz:=3;) wie auch auf die Pro​zeduren (z. B. Zeile 47) zu. Dies funktioniert aber nur dann, wenn dem neu defi​nierten Datentyp TPreis ein Objektname zugeordnet wurde (Z. 43). TPreis ist der Name der Klasse; "p" der des konkreten Exemplars (=Instanz, Objekt) dieser Klasse. Es folgt nun die Erklärung, wie dieses Objekt erzeugt wird. 

Hier sind nochmals die Zeilen 43-45 abgedruckt:

var p: TPreis;

begin

  p:=TPreis.Create;

„Normalerweise“ wird bei der Vereinbarung der Variablen die entsprechende Variable auch im Hauptspeicher angelegt, hier wäre das also: var p: TPreis;

In Delphi werden nun aber Objekte dynamisch erzeugt, das heißt, dass eine spezielle Methode des Objektes aufgerufen wird welche dann das Objekt erst zur Laufzeit erzeugt. Allgemein wird diese Methode als Konstruktor, bezeich​net, in Delphi heißt die Funktion create (s. letzte Zeile im obigen Quellcode). Diese Funktion wird automatisch angelegt. Wird ein Objekt nicht mehr benötigt, kann es mit dem Destruktor wieder aus dem Hauptspeicher entfernt werden (in Pascal/Delphi: Free, vgl Zeile 49: p.Free;). 

Der Objektname (hier: p) ist ebenfalls keine „normale“ Variable, welche die Felder und Methoden des Objektes beinhaltet, sondern eine Referenz (Pointer) auf eine Instanz. Die wird bei Delphi „automatisch“ erledigt, sonst dürfte man nicht p.txt1, sondern p^.txt1. schreiben (weiteres hierzu: s. EILERS, auch BAYER).

Hinweis: In C++ können Objekte auch statisch erzeugt werden; in Java und Pascal/Delphi werden Objekte immer dynamisch angelegt.

Übungsaufgabe: Erstellen Sie ein Programm (objektorientiert) zur Berechnung von Fläche und Umfang eines Kreises!

Lösung (einfach, Radius vorgegeben):

Anlegen der  Klasse TKreis in einer Unit

unit Unit1;

interface

uses sysutils;

type

  TKreis = class

  Radius: double;

  function getUmfang:  double;

  function getFlaeche: double;

end;

implementation

function TKreis.getUmfang:  double;

begin

   getUmfang:=3.1415927*2*Radius;

end;

function TKreis.getFlaeche:  double;

begin

   getFlaeche:=3.1415927*Radius*Radius;

end;

end.

Hauptprogramm:

program Project1;

uses

  SysUtils,   Unit1 in 'Unit1.pas' ;

  {$APPTYPE CONSOLE}

var k: TKreis;

begin

  writeln('Kreisberechnung, Radius = 5');

  k:=TKreis.Create;

  k.Radius := 5;

  writeln('Umfang: ',k.getUmfang);

  writeln('Flaeche: ',k.getFlaeche);

  k.Free;

  readln;

end.

WINDOWS-Programme, Begriffe (nach DOBERENZ, KOWALSKI):

Eigenschaften (Properties): Attribute von Objekten , z. B. Höhe, Breite, Farbe

Methoden: Die in den Objekten definierten Funktionen, sie verändern die Eigenschaften des Objektes

Ereignisse (Events): Aktuelle Nachrichten, die vom Objekt empfangen werden, z. B. Mausklick. Zur Laufzeit des Programms löst ein Ereignis, also z. B. ein Mausklick eine Methode aus, welche dann Eigenschaften des Objektes verändert. Man sagt auch: Die Ereignisbehandlungsroutinen (Methoden, Event-Handler) reagieren dann auf ein spezielles Ereignis.

Windows-Programme, grundsätzliche Verarbeitungsweise: Fenster + Ereignisse (Nachrichten, die vom Objekt empfangen werden), z. B. „Schalter geklickt“) => Reaktion: Ereignisbehandlungsroutine (Event Handler). Dies ist anders als bei der herkömmlichen Programmierung, bei der die Reihenfolge der Abarbeitung der Anweisungen streng vorgeschriebenen ist. Struktogramme können zwar nach wie vor logische Schritte der Routinen verdeutlichen, für die Darstellung eines gesamten Systems sind aber andere, erweiterte Möglichkeiten notwendig (=> UML).

Objekte: insbesondere die Elemente der Windows-Bedienoberfläche, z. B.:

· Formulare (Windows-Fenster), welches weitere Objekte wie Komponenten enthält

· Komponenten wie z. B. Buttons, Edit-Felder usw.

· Sonstiges Objekte wie Drucker oder Windows Zwischenablage

Vorgehensweise:

1. Visueller Entwurf der Oberfläche (z. B. auf das Startformular Buttons platzieren

2. Mit Hilfe des Objektinspektors den Objekten Eigenschaften zuweisen (im einfachsten Fall: Farbe)

3. Ereignisbehandlungsroutinen programmieren (Doppelklick auf die Komponente und vorgefertigten Rahmencode vervollständigen

Objekt und Vererbung

Das folgende Programm generiert sich selbst, wenn nach dem Aufruf von Delphi die F9-Taste gedrückt wird. Die Programmausführung bewirkt, dass das unten stehende Windows-Fenster mit den Standart-Funktionalitäten erscheint (Minimieren, Maximieren, Schließen), ansonsten „kann das Programm nichts“ (man sollte sich trotzdem klarmachen, dass dies schon eine ganze Menge ist). Die Klasse TForm1 hat alle Merkmale ihres Vorfahren TForm geerbt, neue Eigenschaften oder Methoden sind aber nicht hinzugefügt worden.

unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs;

type

  TForm1 = class(TForm)

  private

    { Private-Deklarationen }

  public

    { Public-Deklarationen }

  end;

var

  Form1: TForm1;


implementation

{$R *.DFM}

end.


Die Minimalausstattung eines Windows-Fensters, so wie es von Delphi automatisch erzeugt wird. 

Mit Hilfe z. B. der Komponentenpalette können die Möglichkeiten dieses Objektes nun erweitert werden (Hinzufügen von Komponenten, Methoden und Eigenschaften). Delphi selbst kapselt die komplexen Windows-Oberflächen​elemente in einfach zu handhabenden Komponenten.

Dateien eines Delphi-Projektes

Aufgabe:

Hinzufügen eines ENDE-Buttons in das Formular (hier: Delphi 5):


Folgende Dateien werden angelegt (variiert je nach Version etwas) :


Zunächst die Datei, mit der die ProgrammiererInnen direkt etwas zu tun haben:

Unit1.PAS (die letzte Datei im obigen Explorer-Ausschnitt)

unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

  private

    { Private-Deklarationen }

  public

    { Public-Deklarationen }

  end;

  var

  Form1: TForm1;

implementation

{$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

begin

close;

end;

end.

Hinweis: Der Prozedurname TForm1.Button1Click darf nicht „einfach so“ verändert werden. Der Name ist an anderer Stelle hinterlegt, damit das System „weiß“, welche spezielle Ereignisprozedur abzuarbeiten ist. Eine Namens​änderung „vernichtet“ diese Information. 

Bei der sogenannte „Komponentenentwicklung“ werden neue Komponenten aus bestehenden Delphi Komponenten mittels Vererbung und Ergänzung von neuen Eigenschaften, Methoden und Ereignissen erstellt und in das Delphi-System integriert. Das obige Beispiel könnte bereits in die Komponentenbibliothek gestellt werden. Bei komplexeren Komponenten erfordert dies allerdings relativ viel an Erfahrung und Kenntnissen hinsichtlich der OOP und des Delphi-Sys​tems (vgl. Literatur).

Delphi-Projekt-Datei: Projekt1.DPR
program Project1;

uses

  Forms,

  Unit1 in 'Unit1.pas' {Form1};

{$R *.RES}

begin

  Application.Initialize;

  Application.CreateForm(TForm1, Form1);

  Application.Run;

end.

Hauptprogramm mit begin ... end. Hier erfolgt die Initialisierung inkl. Konstruktoraufruf für Form1 und der eigentliche Start des Programms. So wie auch die übrigen Dateien wird diese Datei automatisch angelegt und sollte nicht manuell verändert werden.

Formular- und Komponenteneinstellungen stehen in der Unit1.dfm-Datei:

object Form1: TForm1

  Left = 192

  Top = 107

  Width = 227

  Height = 167

  Caption = 'Test-Programm'

  Color = clYellow

  Font.Charset = DEFAULT_CHARSET

  Font.Color = clWindowText

  Font.Height = -11

  Font.Name = 'MS Sans Serif'

  Font.Style = []

  OldCreateOrder = False

  PixelsPerInch = 96

  TextHeight = 13

  object Button1: TButton

    Left = 48

    Top = 40

    Width = 121

    Height = 41

    Caption = 'ENDE'

    Font.Charset = DEFAULT_CHARSET

    Font.Color = clRed

    Font.Height = -13

    Font.Name = 'MS Sans Serif'

    Font.Style = [fsBold]

    ParentFont = False

    TabOrder = 0

    OnClick = Button1Click

  end

end

Hinweis:

Alle Klassen sind automatisch Nachkommen einer grundlegenden Basisklasse;

Type TArtikel = class

wird vom Compiler umgesetzt in

Type TArtikel = class(TObject)

Daher hat auch jede Klasse einen Standardkonstruktor (Aufruf mit „create“).

Weitere Dateien:

Projekt1.cfg:
Compileroptionen des Projektes

Projekt1.dof:
Allgemeine Einstellungen der Entwicklungsumgebung

Projekt1.exe:
Lauffähiges Programm

Projekt1.res:
Ressourcendatei für grafische Symbole

Unit1.dcu:

Objektcode der letzten erfolgreichen Übersetzung der Unit1

Unit1.~pas:

.bak-Datei

Jedes Projekt soll (eigentlich: muss) in einem eigenen Projektordnern stehen.

Lösung der Aufgabe zur Preisberechnung in Delphi 6:







Quellcode:

(Hinweis: Dies ist nur eine Lösungsmöglichkeit, es geht auch anders!)



Relationen zwischen Klassen

Einführungsbeispiel

Es soll ein einfaches Würfelspiel programmiert werden: Nach dem Click auf einen Button werden zwei zufällig erzeugte „Würfelaugen“ angezeigt:


Folgende „einfache“ Lösungsmöglichkeit bietet sich zunächst an:

unit Unit1;

interface

.......................{hier: uses}

type

  TForm1 = class(TForm)

    Label1: TLabel;

    Label2: TLabel;

    Button1: TButton;

    procedure FormCreate(Sender: TObject);

    procedure Button1Click(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

var FAugen: byte;

{$R *.dfm}

procedure TForm1.FormCreate(Sender: TObject);

begin

Randomize;

end;

procedure TForm1.Button1Click(Sender: TObject);

begin

  FAugen := Random(6)+1;

  Label1.caption := IntToStr(FAugen);

  FAugen := Random(6)+1;

  Label2.caption := IntToStr(FAugen);

end;

end.

Nachteil dieser Lösung ist die mangelhafte Trennung von GUI (Benutzerober​fläche) und dem „Fachkonzept“ (Würfel werfen). Bei größeren Programmen ist es sinnvoll, Programmteile, die zusammengehören in eigenen Units zu ent​wickeln (Modularisierung). An dem oben vorgeführten Würfelbeispiel werden nun verschiedene Möglichkeiten vorgeführt. Natürlich macht dies erst bei größeren Problemstellungen wirklich Sinn, aber mit dem vorliegende Beispiel lassen sich Relationen zwischen Klassen ohne großen Aufwand verdeutlichen.

Der Würfel als eigene Klasse

Zunächst die GUI:

unit GUI;

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, StdCtrls, Spiel;


type

  TForm1 = class(TForm)

Unit mit dem Würfelobjekt
    Label1: TLabel;

    Label2: TLabel;

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormCreate(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1  : TForm1;

  w1, w2 : TWuerfel;

implementation

 {$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

begin

w1.Werfen;

w2.Werfen;

// Folgendes geht natürlich nicht:

// Label1.caption := IntToStr(w1.FAugen);

// Der Zugriff soll nämlich nur über die Klassenmethoden

// möglich sein!!!

Label1.caption := IntToStr(w1.GetAugen);

Label2.caption := IntToStr(w2.GetAugen);

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

   w1:=TWuerfel.Create;

   w2:=TWuerfel.Create;

   Randomize;

end;

end.

Und hier die Unit mit der Würfel-Klasse

unit Spiel;

interface

type

  TWuerfel = class(TObject)

  private

    FAugen : byte;

  public

    function GetAugen : byte;

    procedure Werfen;

  end;

implementation

function TWuerfel.GetAugen : byte;

begin

  Result := FAugen;

end;

procedure TWuerfel.Werfen;

begin

  FAugen := Random(6)+1;

end;

end.

In dieser Unit befindet sich nur noch das „Fachkonzeptes des Würfels“. Es ist als eigene Klasse mit den notwendigen Methoden und dem Attribut „FAugen“ modelliert.

Vererbung („Ist“-Beziehung)

Es lässt sich leicht vorstellen, dass die Klasse Würfel von einer „Ur-Klasse“ abgeleitet sein könnte, welche unabhängig vom speziellen Würfel mit sechs Augen Zufallszahlen aus einem beliebigen Zahlenraum erzeugt. Dann ließen sich auch Klassen bilden, die z. B. ein Münz-Werfspiel (Kopf oder Zahl: zwei Möglichkeiten) oder ein Glücksrad mit 100 Feldern simulieren. 

Ur-Klasse „Zufallsgerät“

unit Zufallsgeraet;

interface

type

  TZufallsgeraet = class

  private

    anzahl: integer;

    aktuell: integer;

  public

    constructor create(anz: integer);

    procedure aktivieren;

    function getAktuell: integer;

  end;

implementation

constructor TZufallsgeraet.create(anz: integer);

begin

randomize;

anzahl := anz;

aktuell := 0;

end;

procedure TZufallsgeraet.aktivieren;

begin

aktuell :=  random(anzahl) + 1;

end;

function TZufallsgeraet.getAktuell: integer;

begin

result := aktuell;

end;

end.

Das Attribut anzahl der Klasse Zufallsgeraet wird beim Aufruf des Konstruktors initialisiert (Parameterübergabe aus der Unit der abgeleiteten Klasse, welche festlegt, ob es sich um ein Glücksrad, einen Würfel usw. handelt). Der Quellcode der Klasse Zufallsgeraet müsste nicht bekannt sein, lediglich die Namen ihrer Methoden (und falls nötig: Schnittstellen bzw. Rückgabewerte). Die Würfel-Klasse erbt nun das Notwendige der Ur-Klasse: Die aktuell erzeugte Zahl (gewürfelte Augen oder Glücksrad-Zahl) ist in der abgeleiteten Klasse nicht mehr definiert:

unit Spiel;

interface

uses Zufallsgeraet;

type

  TWuerfel = class(TZufallsgeraet)

  public

    constructor create;

    function GetAugen : byte;

    procedure Werfen;

  end;

implementation

constructor TWuerfel.create;

begin

inherited create(6);

end;

function TWuerfel.GetAugen : byte;

begin

  Result := getAktuell;

end;

procedure TWuerfel.Werfen;

begin

  aktivieren;

end;

end.

GUI:

unit GUI;

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, StdCtrls, Spiel;

type

  TForm1 = class(TForm)

    Label1: TLabel;

    Label2: TLabel;

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormCreate(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1  : TForm1;

  w1, w2 : TWuerfel;

implementation

 {$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

begin

w1.Werfen;

w2.Werfen;

Label1.caption := IntToStr(w1.GetAugen);

Label2.caption := IntToStr(w2.GetAugen);

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

   w1:=TWuerfel.Create;

   w2:=TWuerfel.Create;

end;

end.

 „Hat“ (Aggregation) - und „Kennt“ (Assoziation) - Beziehung

Neben der Vererbung existieren noch weiterer Beziehungen zwischen Klassen  (eigentlich: Objekten).

1. Das Spiel-Objekt soll die vollständige Kontrolle über die Würfel haben 

   („Hat“-Beziehung):

„Das Spiel-Objekt soll für Erzeugung und Vernichtung der beiden Würfelob​jekte verantwortlich gemacht werden. Bei Erzeugung des Spiel-Objekts sollen automatisch die Würfel-Objekte mit erzeugt werden. Das kann durch eine Er​weiterung des TSpiel-Konstruktors geschehen. Ebenso soll der Destruktor die Freigabe des Speichers der Würfel-Objekte übernehmen. In einem ersten Schritt wird die Beziehung durch die beiden Referenzvariablen Wuerfel1 und Wuerfel2 in der Klassendefinition von TSpiel zum Ausdruck gebracht“.  (Quelle: s. oben)

Quellcode:

unit uGUI;


interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  uModel, StdCtrls;

type

  TForm1 = class(TForm)

    bWuerfele: TButton;

    lAus1: TLabel;

    lAus2: TLabel;

    procedure FormDestroy(Sender: TObject);

    procedure bWuerfeleClick(Sender: TObject);

    procedure FormCreate(Sender: TObject);

  private

    Spiel1 : TSpiel;

  public

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

procedure TForm1.FormDestroy(Sender: TObject);

begin

  Spiel1.free;

end;

procedure TForm1.bWuerfeleClick(Sender: TObject);

begin

  Spiel1.wuerfeln;

  lAus1.caption := IntToStr(Spiel1.wurf1);

  lAus2.caption := IntToStr(Spiel1.wurf2);

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

  Randomize;

  Spiel1 := TSpiel.Create;

end;

end.

unit uModel;

interface

type

  TWuerfel = class(TObject)

  private

    FAugen : byte;

  public

    procedure SetAugen(a : byte);

    procedure SetRandom;

    function GetAugen : byte;

  end;

  TSpiel = class(TObject)

  private

    Wuerfel1, Wuerfel2 : TWuerfel;

  public

    constructor Create;

    destructor Destroy;

    procedure wuerfeln;

    function wurf1 : byte;

    function wurf2 : byte;

  end;

implementation

procedure TWuerfel.SetAugen(a : byte);

begin

  FAugen := a;

end;

procedure TWuerfel.SetRandom;

begin

  FAugen := Random(6)+1;

end;

function TWuerfel.GetAugen : byte;

begin

  Result := FAugen;

end;

constructor TSpiel.Create;

begin

  inherited Create; // inherited;  hat gleiche Wirkung

  Wuerfel1 := TWuerfel.Create;

  Wuerfel2 := TWuerfel.Create;

end;

destructor TSpiel.Destroy;

begin

  Wuerfel1.Free;

  Wuerfel2.Free;

  inherited Destroy;

end;

procedure TSpiel.wuerfeln;

begin

  Wuerfel1.SetRandom; Wuerfel2.SetRandom;

end;

function TSpiel.wurf1 : byte;

begin

  Result := Wuerfel1.GetAugen;

end;

function TSpiel.wurf2 : byte;

begin

  Result := Wuerfel2.GetAugen;

end;

end.

Hinweis: Vergleiche auch die Beziehung zwischen Form und z. B. den Buttons auf dieser: Die Form „hat“ die Buttons!

Die Komposition ist eine strengere Form der Aggregation. Auf sie wird hier nicht eingegangen. 

 2. Die Würfel-Objekte sind unabhängig vom Spiel-Objekt und führen so ein

„eigenständiges Innenleben“, („Kennt“-Beziehung):

Dem Spiel werden die Würfel über Referenzen bekannt gemacht („kennen“). Eine kleine Erweiterung: Das Spiel zählt die Anzahl der Würfe:

unit uGUI;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  uModel, StdCtrls;

type

  TForm1 = class(TForm)

    bWuerfele: TButton;

    lAus1: TLabel;

    lAus2: TLabel;

    Label1: TLabel;

    Label2: TLabel;

    procedure FormDestroy(Sender: TObject);

    procedure bWuerfeleClick(Sender: TObject);

    procedure FormCreate(Sender: TObject);

  private

    Wuerfel1, Wuerfel2 : TWuerfel;

    Spiel1 : TSpiel;

  public

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

procedure TForm1.FormDestroy(Sender: TObject);

begin

  Spiel1.free;

end;

procedure TForm1.bWuerfeleClick(Sender: TObject);

begin

  label2.caption := IntToStr(Spiel1.wuerfeln);

  lAus1.caption := IntToStr(Spiel1.wurf1);

  lAus2.caption := IntToStr(Spiel1.wurf2);

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

  Randomize;

  Wuerfel1 := TWuerfel.Create;

  Wuerfel2 := TWuerfel.Create;

  Spiel1 := TSpiel.Create(wuerfel1, wuerfel2);

end;

end.

unit uModel;

interface

type

  TWuerfel = class(TObject)

  private

    FAugen : byte;

  public

    procedure SetAugen(a : byte);

    procedure SetRandom;

    function GetAugen : byte;

  end;

  TSpiel = class(TObject)

  private

    wuerfel1: TWuerfel;

    wuerfel2: TWuerfel;

    Anzahl : Integer;

  public

    constructor Create(w1, w2: TWuerfel);

    destructor Destroy;

    function wuerfeln : integer;

    function wurf1 : byte;

    function wurf2 : byte;

  end;

implementation

procedure TWuerfel.SetAugen(a : byte);

begin

  FAugen := a;

end;

procedure TWuerfel.SetRandom;

begin

  FAugen := Random(6)+1;

end;

function TWuerfel.GetAugen : byte;

begin

  Result := FAugen;

end;

constructor TSpiel.Create(w1, w2: TWuerfel);

begin

  inherited Create; // inherited;  hat gleiche Wirkung

  wuerfel1:=w1;

  wuerfel2:=w2;

  Anzahl := 0;

end;

destructor TSpiel.Destroy;

begin

  inherited Destroy;

end;

function TSpiel.wuerfeln:integer;

begin

  Wuerfel1.SetRandom; Wuerfel2.SetRandom;

  Anzahl := Anzahl + 1;

  Result := Anzahl;

end;

function TSpiel.wurf1 : byte;

begin

  Result := Wuerfel1.GetAugen;

end;

function TSpiel.wurf2 : byte;

begin

  Result := Wuerfel2.GetAugen; 

end;

end.

Polymorphismus

Neben den Klassen und der Vererbung ist Polymorphismus der dritte „Grund​pfeiler“ der objektorientierten Programmierung. Polymorphismus bedeutet (u.a.), dass bei abgeleiteten Klassen, welche geerbte Methoden der Elternklasse mit neuem Quellcode überschrieben haben, die gleiche Nachricht ein unter​schiedliches Verhalten bewirkt. Das Senderobjekt der Nachricht muss nicht wissen, zu welcher speziellen abgeleiteten Klasse das Empfängerobjekt gehört; dieses wiederum reagiert ganz individuell auf die Nachricht.

Problemstellung (s. BAYER, Kap. 10):
Ein Programm speichert Daten mehrerer unterschiedlicher, aber ähnlicher Ob​jekte (Kreis, Rechteck, Quadrat). Die einzelnen Objekte (genauer: die Referenz​en dieser Objekte) werden in einem Array gespeichert. Das Array „weiß“ nicht, welche Objekte tatsächlich vorhanden sind, da diese erst während der Laufzeit aufgrund von Benutzereingaben erzeugt werden (die sog. „späte Bindung“). Das Array benötigt aber einen Datentyp. Daher werden Referenzen vom Typ einer Basisklasse, von der alle Grafikklassen abgeleitet sind (Vererbung), im Array gespeichert. Nach den Eingaben der Werte für Flächenberechnungen wird die Gesamtfläche aller Objekte berechnet und ausgegeben. Mit Hilfe von Polymorphismus ist diese Berechnung ohne case- bzw. if-Anweisungen (relativ) einfach zu realisieren (hier als Konsolenanwendung).

unit

unit Unit1;

interface

uses sysutils;

type TGeo = class

  public

    function flaeche: double; virtual; abstract;

end;

Hinweis: Eine virtuelle Methode muss nicht unbedingt abstrakt sein; in der Elternklasse könnte ja trotzdem eine Implementierung dieser Methode vor​liegen. Durch „virtual“ wird festgelegt, dass diese Methode in der Kind-Klasse neu implementiert wird (override). Eine abstrakte Methode ist immer „virtual“; irgendwo muss sie ja dann doch implementiert sein!

type TKreis = class(TGeo)

  private

     radius: double;

  public

     constructor Create(Radius: double);

     function flaeche: double; override;

end;


type TRechteck = class(TGeo)

  private

     laenge: double;

     breite: double;

  public

     constructor Create(laenge, breite: double);

     function flaeche: double; override;

end;

implementation

constructor TKreis.Create(Radius: double);

begin

  inherited Create;

  self.radius := radius;

end;

function TKreis.flaeche: double;

begin

  flaeche:=radius * radius * 3.1415927;

end;

constructor TRechteck.Create(laenge, breite: double);

begin

  inherited Create;

  self.laenge := laenge;

  self.breite := breite;

end;

function TRechteck.flaeche: double;

begin

  flaeche:=laenge * breite;

end;

end.

Hauptprogramm

program Project1;

uses

  SysUtils,

  Unit1 in 'Unit1.pas';

  {$APPTYPE CONSOLE}

var Geo: array [1..3] of TGeo;

    i: integer; ein: string; r,l,b: integer;

    fl_sum:double;

begin

  for i:= 1 to 3 do

  begin

   write('k=Kreis, r=Rechteck: ');

   readln(ein);

   if ein = 'k' then

     begin

       write('Radius: ');

       readln(r);

       Geo[i]:=TKreis.Create(r);

     end

   else

     begin

       write('Laenge: ');

       readln(l);

       write('Breite: ');

       readln(b);

       Geo[i]:=TRechteck.Create(l,b);

     end;

  end;

  fl_sum:=0;

  for i := 1 to 3 do

  begin

    fl_sum:=fl_sum+geo[i].flaeche;

  end;

  writeln;

  writeln('Summe aller Objekte: ',FloatToStr(fl_sum));

  readln;

  for i:=1 to 3 do

  begin

    geo[i].Free;

  end;

end.

Polymorphismus ist das Zusammenwirken von Vererbung, Überschreiben und späte Bindung, eine eigene Syntax (z. B. eine neue Anweisung etc.) zum Polymorphismus existiert nicht.

Programmlauf:


Hier wird nun unter var das vereinbart, was zur Laufzeit des Programms dann zur Instanz  der Klasse wird: das Objekt  Form1.





�





Diese Prozedur wurde quelltextmäßig verändert. Bei Doppelklick auf den Button legte das Delphi-Entwicklungssystem den Prozedurkopf und begin .. end; automatisch an. Die Befehlzeile close; wurde dann von Hand ergänzt. (Ereignisroutine wurde hinzugefügt).





Einige Einstellungen sind hier direkt erkennbar.





program Project1;


uses


  SysUtils,


  Unit1 in 'Unit1.pas';


  {$APPTYPE CONSOLE}


var p: TPreis;


begin


  p:=TPreis.Create;


  p.txt1;


  p.ein_rech_aus;


  p.txt2;


  p.Free;


  readln;


end.

unit Unit1;


interface


uses sysutils;


type


  TPreis = class


  private


   preis: double;


   menge: double;


   g_preis: double;


  public


    proz : integer;


    procedure txt1;


    procedure ein_rech_aus;


    procedure txt2;


end;





implementation


procedure TPreis.txt1();


begin


    writeln('Preisberechnung');


end;





procedure TPreis.ein_rech_aus;


begin


  write('Preis: ');


  readln(preis);


  write('Menge: ');


  readln(menge);


  g_preis:=menge*preis;


  writeln('Gesamtpreis: ',g_preis:8:2);


end;


procedure TPreis.txt2();


begin


  writeln('Programmende');


end;





end.

Destruktor-Aufruf: Löschen der Objekte, Freigabe des Speichers





Hier kommt Polymorphismus ins Spiel. Unabhängig davon, welches Objekt tatsächlich im Array-Element gespeichert ist, kann das Programm die Methode „flaeche“ aufrufen, weil diese in der Basisklasse deklariert ist und in den abgeleiteten Klassen überschrieben wurde. Ohne diesen Mechanismus müsste mit if.. oder case.. gearbeitet werden.





Hier werden die Objekte erzeugt. Der Anwender/die Anwenderin kann hier zwischen Kreis und Rechteck belieb auswählen (das Array „weiß“ nicht, welches Objekt tatsächlich erzeugt wird).





Array vom Typ TGeo (abstrakte Klasse)





Die Methoden zur Berechnung der Flächen





Initialisierung, durch „self“ erfolgt der Bezug auf das Objekt selbst 





Aufruf des (geerbten) Standardkonstruktors (Objekt wird erzeugt).





Die Methode „flaeche“ soll überschrieben werden.





Abgeleitete Klasse von TGeo.





TGeo ist eine abstrakte Basisklasse, da von ihr keine Objekte erzeugt werden können, weil die Methode „flaeche“ nicht implementiert ist. „flaeche“ ist deshalb nicht implementiert, weil das die Nachkommen tun müssen (->unterschiedl. Flächenberechnung von Kreisen und Rechecken). Dies wird mit „abstract“ festegelegt. Damit bei dieser Methode die späte Bindung und das Überschreiben (s. später) funktioniert, ist die Methode  „virtual“.





















































Idee: http://hsg.region-kaiserslautern.de/faecher/inf/material/oop





Methode der Elternklasse





Abgeleitet!





inherited ist hier unbedingt erforderlich (ausprobiern!), s. hierzu das Kapitel „Polymorphismus“


Als Parameter wird die Zahl 6 übergeben: Würfel!





Das Spiel lernt die Würfel kennen, s. unten





Die Würfel werden zentral erzeugt und nicht in „Spiel“ (Eigenleben der Würfel) 





Übergabe der Würfel: Referenzen





Übergabe der Würfel (Referenzen)





Hier werden beide Würfel-Objekte im Spiel-Konstruktor erzeugt, das Spiel „hat“ die Würfel, s. auch Destruktor.





s. unten

Erstellen einer zweiten Unit (hier umbenannt in Preis_Rech)





unit Preis_Rech;





interface





type


  TPreis = class


  private


    preis  : double;


    menge  : integer;


    g_preis: double;


  public


    procedure daten_uebernehmen(preis_ein: double; menge_ein: 


  	 integer);


    function ergebnis: double;


end;





implementation





function TPreis.ergebnis: double;


begin


  g_preis:=menge*preis;


  result := g_preis;


end;


procedure TPreis.daten_uebernehmen(preis_ein: double; menge_ein:


 integer);


begin


    preis := preis_ein;


    menge := menge_ein;


end;


end.





unit Unit1;





interface





uses


  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,


  Dialogs, StdCtrls, Preis_rech;





type


  TForm1 = class(TForm)


    Edit1: TEdit;


    Edit2: TEdit;


    Edit3: TEdit;


    Label1: TLabel;


    Label2: TLabel;


    Label3: TLabel;


    Button1: TButton;


    procedure Button1Click(Sender: TObject);


    procedure FormCreate(Sender: TObject);


  private


    preis : TPreis;


  public


    { Public-Deklarationen }


  end;





var


  Form1: TForm1;


  code             : integer;


  einzel_preis_ein : double;


  menge_ein        : integer;


  g_preis_aus      : double;


  p_aus            : string;


implementation





{$R *.dfm}





procedure TForm1.FormCreate(Sender: TObject);


begin


  preis := TPreis.Create;


end;





procedure TForm1.Button1Click(Sender: TObject);


begin


val(edit1.Text,einzel_preis_ein, code);


val(edit2.Text,menge_ein, code);





preis.daten_uebernehmen(einzel_preis_ein, menge_ein);


g_preis_aus:=preis.ergebnis;





str(g_preis_aus:8:2, p_aus);


edit3.Text:= p_aus;





end;





end.

